
MMeenngggguunnaakkaann  DDrraagg--DDrroopp  
 

14.1. Apa Itu Drag-Drop? 
 
Drag-Drop merupakan istilah umum di dalam penggunaan mouse untuk menggeser, menyalin atau 
memindahkan gambar, teks,  file, dll. Menggunakan drag-drop akan mempermudah user saat 
menggunakan sebuah program. 
 
14.2. Drag-Drop dengan VB 6.0 
 
VB 6.0 menyediakan beberapa property, method dan event yang berhubungan dengan drag-drop. 
Operasi drag-drop melibatkan object source dan object target. Setiap object di dalam form bisa menjadi 
source ataupun target (termasuk form itu sendiri). 

Property, method dan event yang berhubungan dengan drag-drop adalah sebagai berikut: 
 

Property Object Keterangan 
DragMode Source Menentukan bagaimana drag-drop dimulai: 

- Manual (vbManual = 0, default) atau 
- Otomatis (vbAutomatic = 1) 

DragIcon Source Menentukan bentuk pointer saat operasi drag-drop 
berlangsung. 

Method Object Keterangan 
Drag [action] Source Digunakan untuk memulai drag-drop secara manual 

(property DragMode diset vbManual). 
Event Object Keterangan 
DragOver Target Saat object source di-drag melewati object target. 
DragDrop Target Saat object source di-drop pada object target. 

 
Nilai untuk argumen action adalah sebagai berikut: 
 

Nilai Keterangan 
vbBeginDrag = 1 Memulai operasi drag-drop 
vbCancel = 0 Membatalkan operasi drag-drop 
vbEndDrag = 2 Mengakhiri operasi drag-drop 

 
Ada tiga argumen pada event-event drag-drop, yaitu: Source, X, Y dan State. Argumen Source 
menunjukkan object yang menjadi source. Argumen X dan Y menunjukkan posisi koordinat pointer 
mouse. Sedangkan argumen State menunjukkan status pointer pada saat event DragOver, nilainya terdiri 
dari: 0  pointer masuk ke dalam object target, 1  pointer meninggalkan object target dan 2  pointer 
bergerak di dalam object target. 
 
14.3. Contoh Program: Drag-Drop Test 
 
1. Aktifkan program Microsoft Visual Basic 6.0. 
2. Bukalah kembali file project Latihan.vbp. 
3. Tambahkan form baru ke dalam project. 
4. Pada Jendela Object buatlah User Interface seperti ini: 

Setting property object-nya adalah sebagai berikut: 
 

Object Properties Value 
Form14 BorderStyle 

Caption 
StartUpPosition 

1 – Fixed Single 
Drag and Drop Test 
2 – Center Screen 

Text1 DragMode 
Text 

1 – Automatic 
<kosong> 

 
5. Buka Jendela Code, lalu ketikkan kode programnya seperti pada Listing-14A. 
6. Simpanlah Form14 dengan nama file Lat14.frm. 

Catatan: 
Anda bisa men-download file vb6-latihan.zip. 

7. Klik menu Project > Project1 Properties lalu klik tab General. Gantilah Startup Object-nya menjadi 
Form14. 

8. Coba jalankan project-nya: 
 

 

Keterangan: 
1. Ketikkan sesuatu pada Text1. 
2. Lakukan drag-drop ke List1. 
3. Perhatikan nilai argumen pada event 

DragOver. 

 
9. Ubahlah property object Text1 sebagai berikut: 

 
Object Properties 

Text1 DragIcon 
DragMode 

 
10. Modifikasi kode programnya seperti pada Listing-14B. 
11. Coba jalankan kembali project-nya. 

 
*** 

Listing-14A 
 
Private Sub List1_DragOver(Source As Control, X As Single, Y As Single, _ 
                           State As Integer) 
    Me.Cls 
    Me.Print "Source: " & Source.Name 
    Me.Print "X,Y: " & X & "," & Y 
    Me.Print "State: " & State 
End Sub 
 
Private Sub List1_DragDrop(Source As Control, X As Single, Y As Single) 
    If Source.Name = "Text1" Then 
        List1.AddItem Text1.Text 
    End If 
    Me.Cls 
End Sub 
 
Private Sub Form_DragOver(Source As Control, X As Single, Y As Single, _ 
                          State As Integer) 
    Me.Cls 
End Sub 
 
Private Sub Form_DragDrop(Source As Control, X As Single, Y As Single) 
    Me.Cls 
End Sub 

Listing-14B 
 
Private Sub List1_DragOver(Source As Control, X As Single, Y As Single, _ 
                           State As Integer) 
    Me.Cls 
    Me.Print "Source: " & Source.Name 
    Me.Print "X,Y: " & X & "," & Y 
    Me.Print "State: " & State 
    If Source.Name = "Text1" And State = 0 Then 
        Text1.DragIcon = LoadPicture(App.Path & "DropYes.cur") 
    End If 
End Sub 
 
Private Sub List1_DragDrop(Source As Control, X As Single, Y As Single) 
    If Source.Name = "Text1" Then 
        List1.AddItem Text1.Text 
        Text1.Drag vbEndDrag 
    End If 
    Me.Cls 
End Sub 
 
Private Sub Form_DragOver(Source As Control, X As Single, Y As Single, _ 
                          State As Integer) 
    Me.Cls 
    If Source.Name = "Text1" And State = 0 Then 
        Text1.DragIcon = LoadPicture(App.Path & "DropNo.cur") 
    End If 
End Sub 
 
Private Sub Form_DragDrop(Source As Control, X As Single, Y As Single) 
    If Source.Name = "Text1" Then 
        Text1.Drag vbCancel 
    End If 
    Me.Cls 
End Sub 
 
Private Sub Text1_MouseDown(Button As Integer, Shift As Integer, _ 
                            X As Single, Y As Single) 
    If Text1.Text <> "" And Button = vbLeftButton Then 
        Text1.Drag vbBeginDrag 
    End If 
End Sub 




