

Cepat Mahir Visual Basic 6.0

Krisnha D. Oktovhiana
mail4krisna@yahoo.com

Lisensi Dokumen:

Copyright © 2005 IlmuKomputer.Com

*Seluruh dokumen di **IlmuKomputer.Com** dapat digunakan, dimodifikasi dan disebarkan secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin terlebih dahulu dari **IlmuKomputer.Com**.*

Bab 14

Menggunakan Drag-Drop

14.1. Apa Itu Drag-Drop?

Drag-Drop merupakan istilah umum di dalam penggunaan mouse untuk menggeser, menyalin atau memindahkan gambar, teks, file, dll. Menggunakan *drag-drop* akan mempermudah *user* saat menggunakan sebuah program.

14.2. Drag-Drop dengan VB 6.0

VB 6.0 menyediakan beberapa property, method dan event yang berhubungan dengan *drag-drop*. Operasi *drag-drop* melibatkan object source dan object target. Setiap object di dalam form bisa menjadi source ataupun target (termasuk form itu sendiri).

Property, method dan event yang berhubungan dengan *drag-drop* adalah sebagai berikut:

Property	Object	Keterangan
DragMode	Source	Menentukan bagaimana <i>drag-drop</i> dimulai: - Manual (vbManual = 0, default) atau - Otomatis (vbAutomatic = 1)
DragIcon	Source	Menentukan bentuk pointer saat operasi <i>drag-drop</i> berlangsung.
Method	Object	Keterangan
Drag [action]	Source	Digunakan untuk memulai <i>drag-drop</i> secara manual (property DragMode diset vbManual).
Event	Object	Keterangan
DragOver	Target	Saat object source di- <i>drag</i> melewati object target.
DragDrop	Target	Saat object source di-drop pada object target.


Nilai untuk argumen action adalah sebagai berikut:

Nilai	Keterangan
vbBeginDrag = 1	Memulai operasi <i>drag-drop</i>
vbCancel = 0	Membatalkan operasi <i>drag-drop</i>
vbEndDrag = 2	Mengakhiri operasi <i>drag-drop</i>

Ada tiga argumen pada event-event *drag-drop*, yaitu: **Source**, **X**, **Y** dan **State**. Argumen Source menunjukkan object yang menjadi source. Argumen X dan Y menunjukkan posisi koordinat pointer mouse. Sedangkan argumen State menunjukkan status pointer pada saat event DragOver, nilainya terdiri dari: 0 → pointer masuk ke dalam object target, 1 → pointer meninggalkan object target dan 2 → pointer bergerak di dalam object target.

14.3. Contoh Program: Drag-Drop Test


1. Aktifkan program Microsoft Visual Basic 6.0.
2. Bukalah kembali file project **Latihan.vbp**.
3. Tambahkan form baru ke dalam project.
4. Pada Jendela Object buatlah *User Interface* seperti ini:



Setting property object-nya adalah sebagai berikut:

Object	Properties	Value
Form14	BorderStyle Caption StartPosition	1 – Fixed Single Drag and Drop Test 2 – Center Screen
Text1	DragMode Text	1 – Automatic <kosong>

- Buka Jendela Code, lalu ketikkan kode programnya seperti pada Listing-14A.
- Simpanlah Form14 dengan nama file Lat14.frm.
Catatan:
Anda bisa men-download file vb6-latihan.zip.
- Klik menu Project > Project1 Properties lalu klik tab General. Gantilah Startup Object-nya menjadi Form14.
- Coba jalankan project-nya:



Keterangan:

- Ketikkan sesuatu pada Text1.
- Lakukan *drag-drop* ke List1.
- Perhatikan nilai argumen pada event DragOver.

- Ubahlah property object Text1 sebagai berikut:

Object	Properties
Text1	DragIcon DragMode

- Modifikasi kode programnya seperti pada Listing-14B.
- Coba jalankan kembali project-nya.

Listing-14A

```
Private Sub List1_DragOver(Source As Control, X As Single, Y As Single, _
    State As Integer)
    Me.Cls
    Me.Print "Source: " & Source.Name
    Me.Print "X,Y: " & X & ", " & Y
    Me.Print "State: " & State
End Sub

Private Sub List1_DragDrop(Source As Control, X As Single, Y As Single)
    If Source.Name = "Text1" Then
        List1.AddItem Text1.Text
    End If
    Me.Cls
End Sub

Private Sub Form_DragOver(Source As Control, X As Single, Y As Single, _
    State As Integer)
    Me.Cls
End Sub

Private Sub Form_DragDrop(Source As Control, X As Single, Y As Single)
    Me.Cls
End Sub
```

Listing-14B

```
Private Sub List1_DragOver(Source As Control, X As Single, Y As Single, _
                           State As Integer)
    Me.Cls
    Me.Print "Source: " & Source.Name
    Me.Print "X,Y: " & X & ", " & Y
    Me.Print "State: " & State
    If Source.Name = "Text1" And State = 0 Then
        Text1.DragIcon = LoadPicture(App.Path & "DropYes.cur")
    End If
End Sub

Private Sub List1_DragDrop(Source As Control, X As Single, Y As Single)
    If Source.Name = "Text1" Then
        List1.AddItem Text1.Text
        Text1.Drag vbEndDrag
    End If
    Me.Cls
End Sub

Private Sub Form_DragOver(Source As Control, X As Single, Y As Single, _
                           State As Integer)
    Me.Cls
    If Source.Name = "Text1" And State = 0 Then
        Text1.DragIcon = LoadPicture(App.Path & "DropNo.cur")
    End If
End Sub

Private Sub Form_DragDrop(Source As Control, X As Single, Y As Single)
    If Source.Name = "Text1" Then
        Text1.Drag vbCancel
    End If
    Me.Cls
End Sub

Private Sub Text1_MouseDown(Button As Integer, Shift As Integer, _
                             X As Single, Y As Single)
    If Text1.Text <> "" And Button = vbLeftButton Then
        Text1.Drag vbBeginDrag
    End If
End Sub
```